Find Inbound Idocs for a particular PO number

Covers ORDRSP, SHPCON, MBGMCR, INVOIC
Author: Kevin Wilson

1 Program
1.1 Main report

1.1.1 ZMM_EDI_GETIDOCSFROMPO

*&---------------------------------------------------------------------*
*& Report  ZMM_EDI_GETIDOCSFROMPO
*&---------------------------------------------------------------------*
*& AUTHOR     : Kevin Wilson
*& DESCRIPTION: Report that takes a PO number and searches for IDocs linked
*&              to that PO.
*&---------------------------------------------------------------------*

REPORT  zmm_edi_getidocsfrompo.

*** DATA DEFINITIONS ***
INCLUDE zmm_edi_getidocsfrompo_data.

*** Selection Screen ***
SELECTION-SCREEN BEGIN OF BLOCK s1 WITH FRAME TITLE text-bl1.
PARAMETERS: p_po LIKE ekko-ebeln OBLIGATORY.
SELECTION-SCREEN END OF BLOCK s1.

SELECTION-SCREEN BEGIN OF BLOCK s3 WITH FRAME TITLE text-bl3.
SELECT-OPTIONS so_dat FOR edidc-credat OBLIGATORY.
SELECTION-SCREEN END OF BLOCK s3.

SELECTION-SCREEN BEGIN OF BLOCK s2 WITH FRAME TITLE text-bl2.
PARAMETERS: p_ordrsp TYPE c AS CHECKBOX,
            p_mbgmcr TYPE c AS CHECKBOX,
            p_shpcon TYPE c AS CHECKBOX,
            p_invoic TYPE c AS CHECKBOX.
SELECTION-SCREEN END OF BLOCK s2.

SELECTION-SCREEN BEGIN OF BLOCK s4 WITH FRAME TITLE text-bl4.
PARAMETERS: p_zeros TYPE c AS CHECKBOX DEFAULT 'X',
            p_we09  TYPE c AS CHECKBOX MODIF ID we9.
SELECTION-SCREEN END OF BLOCK s4.

*** INITIALIZATION ***
INITIALIZATION.

* check authority
  AUTHORITY-CHECK   OBJECT 'S_IDOCMONI'
        ID 'EDI_TCD' FIELD 'WE09'
        ID 'ACTVT'   FIELD '03'
        ID 'EDI_DIR' DUMMY
        ID 'EDI_MES' DUMMY
        ID 'EDI_PRN' DUMMY
        ID 'EDI_PRT' DUMMY.
  IF NOT sy-subrc IS INITIAL.
* authority check negative; message for the user and stop program

    LOOP AT SCREEN.
      IF screen-group1 = 'WE9'.
        screen-invisible = '1'.
        MODIFY SCREEN.
      ENDIF.
    ENDLOOP.

  ENDIF.

  p_mbgmcr = 'X'.
  p_shpcon = 'X'.
  p_invoic = 'X'.
  p_ordrsp = 'X'.
  so_dat-sign   = 'I'.
  so_dat-option = 'BT'.
  so_dat-low    = sy-datum - 2.
  so_dat-high   = sy-datum.
  APPEND so_dat.

*** SELECTION ***
START-OF-SELECTION.

  IF p_we09 = 'X'.
    MESSAGE i999(b1) WITH text-m01.
  ENDIF.

  REFRESH: itab_selection,
           int_edidc,
           int_edidd.

  gv_po = p_po.
  IF p_zeros = 'X'.
    SHIFT gv_po LEFT DELETING LEADING '0'.
  ENDIF.

  SELECT SINGLE ebeln INTO ekko-ebeln
    FROM ekko
    WHERE ebeln = p_po.

  IF sy-subrc <> 0.   "No such PO
    MESSAGE i999(b1) WITH text-er1 p_po.
  ELSE.
    gv_execute = 'X'.

    LOOP AT so_dat.

      wa_selection-selname = 'CREDAT'.
      wa_selection-kind    = 'S'.
      MOVE-CORRESPONDING so_dat TO wa_selection.
      APPEND wa_selection TO itab_selection.

    ENDLOOP.

    CLEAR: wa_selection.

* Message Type
    wa_selection-sign    = 'I'.
    wa_selection-option  = 'EQ'.

** ORDRSP
    IF p_ordrsp = 'X'.
* Message Type
      wa_selection-selname = 'MESTYP'.
      wa_selection-kind    = 'S'.
      wa_selection-low     = 'ORDRSP'.
      APPEND wa_selection TO itab_selection.
* Segment containing PO number
      wa_selection-selname = 'SEGMENT1'.
      wa_selection-kind    = 'P'.
      wa_selection-low     = c_segnam_e1edk02.
      APPEND wa_selection TO itab_selection.
* Field Containing PO number
      wa_selection-selname = 'FIELD1_1'.
      wa_selection-kind    = 'P'.
      wa_selection-low     = c_field_belnr.
      APPEND wa_selection TO itab_selection.
* Direction
      wa_selection-selname = 'DIRECT'.
      wa_selection-kind    = 'S'.
      wa_selection-sign    = 'I'.
      wa_selection-option  = 'EQ'.
      wa_selection-low     = '2'.
      APPEND wa_selection TO itab_selection.
* PO Number
      wa_selection-selname = 'VALUE1_1'.
      wa_selection-kind    = 'P'.
      wa_selection-low     = gv_po.
      APPEND wa_selection TO itab_selection.

** Retrieve IDoc control records for selection
      SELECT *
        FROM edidc
        APPENDING TABLE int_edidc
        WHERE credat IN so_dat AND
              mestyp = 'ORDRSP' AND
*             mescod in so_cod and
              direct = '2'.

    ENDIF.

** MBGMCR
    IF p_mbgmcr = 'X'.
* Message Type
      wa_selection-selname = 'MESTYP'.
      wa_selection-kind    = 'S'.
      wa_selection-low     = 'MBGMCR'.
      APPEND wa_selection TO itab_selection.
* Segment containing PO number
      wa_selection-selname = 'SEGMENT1'.
      wa_selection-kind    = 'P'.
      wa_selection-low     = c_segnam_e1bp2017.
      APPEND wa_selection TO itab_selection.
* Field Containing PO number
      wa_selection-selname = 'FIELD1_1'.
      wa_selection-kind    = 'P'.
      wa_selection-low     = c_field_po_number.
      APPEND wa_selection TO itab_selection.
* Direction
      wa_selection-selname = 'DIRECT'.
      wa_selection-kind    = 'S'.
      wa_selection-sign    = 'I'.
      wa_selection-option  = 'EQ'.
      wa_selection-low     = '2'.
      APPEND wa_selection TO itab_selection.
* PO Number
      wa_selection-selname = 'VALUE1_1'.
      wa_selection-kind    = 'P'.
      wa_selection-low     = gv_po.
      APPEND wa_selection TO itab_selection.

** Retrieve IDoc control records for selection
      SELECT *
        FROM edidc
        APPENDING TABLE int_edidc
        WHERE credat IN so_dat AND
              mestyp = 'MBGMCR' AND
*             mescod in so_cod and
              direct = '2'.

    ENDIF.

** SHPCON
    IF p_shpcon = 'X'.
* Message Type
      wa_selection-selname = 'MESTYP'.
      wa_selection-kind    = 'S'.
      wa_selection-low     = 'SHPCON'.
      APPEND wa_selection TO itab_selection.
* Segment containing PO number
      wa_selection-selname = 'SEGMENT1'.
      wa_selection-kind    = 'P'.
      wa_selection-low     = c_segnam_e1edl41.
      APPEND wa_selection TO itab_selection.
* Field Containing PO number
      wa_selection-selname = 'FIELD1_1'.
      wa_selection-kind    = 'P'.
      wa_selection-low     = c_field_bstnr.
      APPEND wa_selection TO itab_selection.
* Direction
      wa_selection-selname = 'DIRECT'.
      wa_selection-kind    = 'S'.
      wa_selection-sign    = 'I'.
      wa_selection-option  = 'EQ'.
      wa_selection-low     = '2'.
      APPEND wa_selection TO itab_selection.
* PO Number
      wa_selection-selname = 'VALUE1_1'.
      wa_selection-kind    = 'P'.
      wa_selection-low     = gv_po.
      APPEND wa_selection TO itab_selection.

** Retrieve IDoc control records for selection
      SELECT *
        FROM edidc
        APPENDING TABLE int_edidc
        WHERE credat IN so_dat AND
              mestyp = 'SHPCON' AND
*             mescod in so_cod and
              direct = '2'.
    ENDIF.

** INVOIC
    IF p_invoic = 'X'.
* Message Type
      wa_selection-selname = 'MESTYP'.
      wa_selection-kind    = 'S'.
      wa_selection-low     = 'INVOIC'.
      APPEND wa_selection TO itab_selection.
* Segment containing PO number
      wa_selection-selname = 'SEGMENT1'.
      wa_selection-kind    = 'P'.
      wa_selection-low     = c_segnam_e1edp02.
      APPEND wa_selection TO itab_selection.
* Field Containing PO number
      wa_selection-selname = 'FIELD1_1'.
      wa_selection-kind    = 'P'.
      wa_selection-low     = c_field_belnr.
      APPEND wa_selection TO itab_selection.
* Direction
      wa_selection-selname = 'DIRECT'.
      wa_selection-kind    = 'S'.
      wa_selection-sign    = 'I'.
      wa_selection-option  = 'EQ'.
      wa_selection-low     = '2'.
      APPEND wa_selection TO itab_selection.
* PO Number
      wa_selection-selname = 'VALUE1_1'.
      wa_selection-kind    = 'P'.
      wa_selection-low     = gv_po.
      APPEND wa_selection TO itab_selection.

** Retrieve IDoc control records for selection
      SELECT *
        FROM edidc
        APPENDING TABLE int_edidc
        WHERE credat IN so_dat AND
              mestyp = 'INVOIC' AND
*             mescod in so_cod and
              direct = '2'.
    ENDIF.

  ENDIF.

  DESCRIBE TABLE int_edidc LINES gv_data_records.

  LOOP AT itab_selection INTO wa_selection.

    CASE wa_selection-selname.

      WHEN 'MESTYP'.
        gv_mestyp = wa_selection-low.

      WHEN 'SEGMENT1'.
        gv_segment = wa_selection-low.

      WHEN 'FIELD1_1'.
        gv_field   = wa_selection-low.

      WHEN 'VALUE1_1'.
        gv_value   = wa_selection-low.

* read information about segment-structure
        PERFORM prepare_datarecord_search.

        LOOP AT int_edidc WHERE mestyp = gv_mestyp.

          REFRESH int_edidd.

          CALL FUNCTION 'IDOC_READ_COMPLETELY'
            EXPORTING
              document_number         = int_edidc-docnum
            TABLES
              int_edidd               = int_edidd
            EXCEPTIONS
              document_not_exist      = 1
              document_number_invalid = 2
              OTHERS                  = 3.
          IF sy-subrc <> 0.
            MESSAGE i999(b1) WITH text-er3 int_edidc-docnum.
          ENDIF.

          PERFORM search_datarecord.

* if no value is found take IDoc anyway
          IF gv_found <> 'X'.
            DELETE int_edidc.
          ELSE.
* We store the segment where the number was found in the MaxSegnum field
            int_edidc-maxsegnum = gv_segnum.
            MODIFY int_edidc.

            CLEAR: gv_segnum, gv_found.
          ENDIF.

        ENDLOOP.

    ENDCASE.

  ENDLOOP.


END-OF-SELECTION.

*** MAIN PROGRAM ***
  IF gv_execute = 'X'.
    IF p_we09 = 'X'.
      SUBMIT rseidoc9 WITH SELECTION-TABLE itab_selection AND RETURN.
    ELSE.

      IF NOT int_edidc[] IS INITIAL.
        DESCRIBE TABLE int_edidc LINES gv_data_records_found.
        MESSAGE s999(b1) WITH text-i01 gv_data_records_found.
        CALL SCREEN 2000.
      ELSE.
        MESSAGE i999(b1) WITH text-er4 gv_data_records.
      ENDIF.

    ENDIF.

  ENDIF.

  INCLUDE zmm_edi_getidocsfrompo_prepf01.

  INCLUDE zmm_edi_getidocsfrompo_stato01.

  INCLUDE zmm_edi_getidocsfrompo_useri01.
1.2 Includes

1.2.1 ZMM_EDI_GETIDOCSFROMPO_DATA

*&---------------------------------------------------------------------*
*&  Include           ZMM_EDI_GETIDOCSFROMPO_DATA
*&---------------------------------------------------------------------*

*** TABLES ***
TABLES: ekko,
        edidc.

*** CONSTANTS ***
CONSTANTS: c_segnam_e1edk02  LIKE edisyn-segtyp        VALUE 'E1EDK02',
           c_segnam_e1edp02  LIKE edisyn-segtyp        VALUE 'E1EDP02',
           c_segnam_e1bp2017 LIKE edisyn-segtyp        VALUE 'E1BP2017_GM_ITEM_CREATE',
           c_segnam_e1edl41  LIKE edisyn-segtyp        VALUE 'E1EDL41',
           c_mestyp_invoic   LIKE edidc-mestyp         VALUE 'INVOIC',
           c_mestyp_mbgmcr   LIKE edidc-mestyp         VALUE 'MBGMCR',
           c_mestyp_shpcon   LIKE edidc-mestyp         VALUE 'SHPCON',
           c_mestyp_ordrsp   LIKE edidc-mestyp         VALUE 'ORDRSP',
           c_field_belnr     LIKE edipar0002-edi_field VALUE 'BELNR',
           c_field_bstnr     LIKE edipar0002-edi_field VALUE 'BSTNR',
           c_field_po_number LIKE edipar0002-edi_field VALUE 'PO_NUMBER',
           c_container(40) VALUE 'CUSTOM_CONTAINER',
           c_itab_alv TYPE tabname VALUE 'EDIDC'.

*** INTERNAL TABLES AND WORKAREAS ***
DATA: itab_selection TYPE STANDARD TABLE OF rsparams,
      wa_selection   TYPE rsparams,
      int_edidc TYPE STANDARD TABLE OF edidc WITH KEY docnum
                    WITH HEADER LINE,
      int_edidd TYPE STANDARD TABLE OF edidd WITH HEADER LINE,
      itab_return TYPE STANDARD TABLE OF bapireturn.

*** GLOBAL VARIABLES ***
DATA: gv_execute(1) TYPE c,
      gv_po         LIKE edipar0002-edi_value,
      gv_data_records type i,
      gv_data_records_found type i,
      gv_segment    LIKE edisegmhd-segtyp,
      gv_mestyp     LIKE edidc-mestyp,
      gv_field      LIKE edipar0002-edi_field,
      gv_value      LIKE edipar0002-edi_value,
      gv_found      TYPE boole_d,
      gv_segnum     like edidd-segnum,
      okcode        LIKE sy-ucomm.

DATA: expleng1_1 TYPE i, offset1_1 TYPE syfdpos,
      expleng1_2 TYPE i, offset1_2 TYPE syfdpos.

*** GRID
* Variables
FIELD-SYMBOLS: <fs_val> TYPE ANY.

DATA: gt_fldcat       TYPE lvc_t_fcat,
      gs_variant      TYPE disvariant.      "for parameter IS_VARIANT

*---------------------------------------------------------------------*
*       CLASS lcl_events DEFINITION
*---------------------------------------------------------------------*
CLASS lcl_events DEFINITION.

  PUBLIC SECTION.

    METHODS:
      on_link_click FOR EVENT hotspot_click
                        OF cl_gui_alv_grid
                        IMPORTING e_row_id
                                  e_column_id
                                  es_row_no,
      handle_double_click
        FOR EVENT double_click OF cl_gui_alv_grid
            IMPORTING e_row e_column.

ENDCLASS.                    "lcl_events DEFINITION


DATA: gref_container  TYPE REF TO cl_gui_custom_container,
      gref_grid       TYPE REF TO cl_gui_alv_grid,
      gref_events     TYPE REF TO lcl_events.

*---------------------------------------------------------------------*
*       CLASS lcl_events  IMPLIMENTATION
*---------------------------------------------------------------------*
CLASS lcl_events IMPLEMENTATION.

  METHOD on_link_click.

    DATA: wa_edidc TYPE edidc,
          t_docnum LIKE edidc-docnum,
          t1.

    CHECK e_row_id GT 0.
    READ TABLE int_edidc INTO wa_edidc INDEX e_row_id.

    IF sy-subrc EQ 0.

      ASSIGN COMPONENT e_column_id OF STRUCTURE wa_edidc
        TO <fs_val>.

      IF sy-subrc EQ 0.

        CASE e_column_id.

          WHEN 'DOCNUM'.

            t_docnum = <fs_val>.

            CASE wa_edidc-mestyp.

              WHEN c_mestyp_invoic.

                CALL DIALOG 'Z_IDOC_INVOIC'
                  EXPORTING
                    display      FROM 'X'
                    docnum       FROM t_docnum
*                    menu_mode    FROM menu_mode
*                    PO_INFO      from po_info
                  IMPORTING
                    return       TO itab_return.

              WHEN c_mestyp_mbgmcr.

                CALL FUNCTION 'ZCA_IDOC_DISPLAY_MBGMCR_RFC'
                  STARTING NEW TASK t1
                  EXPORTING
                    docnum              = t_docnum
                    display             = 'A'
                  EXCEPTIONS
                    idoc_does_not_exist = 1
                    OTHERS              = 2.

                IF sy-subrc <> 0.
                  MESSAGE i999(b1) WITH text-er3 t_docnum.
                ENDIF.

              WHEN c_mestyp_shpcon.

                CALL FUNCTION 'ZCA_IDOC_DISPLAY_DRPSHP_RFC'
                  STARTING NEW TASK t1
                  EXPORTING
                    docnum              = t_docnum
                    display             = 'A'
                  EXCEPTIONS
                    idoc_does_not_exist = 1
                    OTHERS              = 2.
                IF sy-subrc <> 0.
                  MESSAGE i999(b1) WITH text-er3 t_docnum.
                ENDIF.

              WHEN c_mestyp_ordrsp.

                CALL DIALOG 'ZCA_IDOC_ORDRSP'
                  EXPORTING
                    display      FROM 'X'
                    docnum       FROM t_docnum
*                    shortmessage FROM shortmessage
                  IMPORTING
                    return       TO itab_return.

            ENDCASE.

        ENDCASE.

      ENDIF.

    ENDIF.

    CLEAR okcode.

  ENDMETHOD.                    "on_link_click

  METHOD handle_double_click.

    DATA: docnum LIKE edidc-docnum,
          range_credat LIKE RANGE OF edidc-credat,
          wa_edidc LIKE edidc.

    IF okcode IS INITIAL.

* read selected row from internal table
      READ TABLE int_edidc INTO wa_edidc INDEX e_row-index.

      IF sy-subrc = 0.
        docnum = wa_edidc-docnum.
        IF NOT docnum IS INITIAL.
* init range_credat
          CLEAR range_credat.
          REFRESH range_credat.
* submit report in order to display the IDoc
          SUBMIT rseidoc2 WITH docnum = docnum
                          WITH credat IN range_credat
                          AND RETURN.
        ENDIF.

      ENDIF.

    ENDIF.

    CLEAR okcode.

  ENDMETHOD.                           "handle_double_click

ENDCLASS.                    "lcl_events IMPLEMENTATION

1.2.2 ZMM_EDI_GETIDOCSFROMPO_PREPF01
*----------------------------------------------------------------------*
***INCLUDE ZMM_EDI_GETIDOCSFROMPO_PREPF01 .
*----------------------------------------------------------------------*
*&---------------------------------------------------------------------*
*&      Form  prepare_datarecord_search
*&---------------------------------------------------------------------*
FORM prepare_datarecord_search .

  DATA: int_edisegstru TYPE STANDARD TABLE OF edisegstru
                           WITH HEADER LINE.

  CLEAR: offset1_1.
  expleng1_1 = 1000.

  CALL FUNCTION 'SEGMENT_READ'
    EXPORTING
      segmenttyp           = gv_segment
    TABLES
      segmentstructure     = int_edisegstru
    EXCEPTIONS
      no_authority         = 1
      segment_not_existing = 2
      OTHERS               = 3.
  IF sy-subrc IS NOT INITIAL.
    MESSAGE i999(b1) WITH text-er2 gv_segment.
    EXIT.
  ENDIF.

  LOOP AT int_edisegstru.
    IF int_edisegstru-fieldname = gv_field.
      expleng1_1 = int_edisegstru-expleng.
      EXIT.
    ENDIF.
    ADD int_edisegstru-expleng TO offset1_1.
  ENDLOOP.

ENDFORM.                    " prepare_datarecord_search
*&---------------------------------------------------------------------*
*&      Form  search_datarecord
*&---------------------------------------------------------------------*
FORM search_datarecord .

  CLEAR: gv_found.

* Search in specified segments of data record.
  LOOP AT int_edidd WHERE segnam = gv_segment.
    FIND gv_value IN SECTION OFFSET offset1_1 LENGTH expleng1_1
        OF int_edidd-sdata IGNORING CASE.
    IF sy-subrc IS INITIAL.
      gv_found = 'X'.
      gv_segnum = int_edidd-segnum.
      EXIT.
    ENDIF.
  ENDLOOP.

ENDFORM.                    " search_datarecord
*&---------------------------------------------------------------------*
*&      Form  write_report
*&---------------------------------------------------------------------*
FORM write_report  TABLES   p_edidc STRUCTURE edidc.

  LOOP AT p_edidc.

    WRITE: /1 p_edidc-docnum, p_edidc-mestyp, p_edidc-mescod, p_edidc-credat, p_edidc-cretim, p_edidc-sndprn.

  ENDLOOP.

ENDFORM.                    " write_report
*&---------------------------------------------------------------------*
*&      Form  field_catalog_merge
*&---------------------------------------------------------------------*
*----------------------------------------------------------------------*
FORM field_catalog_merge .

  FIELD-SYMBOLS: <fs_fldcat> TYPE lvc_s_fcat.

  CALL FUNCTION 'LVC_FIELDCATALOG_MERGE'
    EXPORTING
      i_structure_name       = c_itab_alv
    CHANGING
      ct_fieldcat            = gt_fldcat
    EXCEPTIONS
      inconsistent_interface = 0
      program_error          = 0
      OTHERS                 = 0.

  LOOP AT gt_fldcat ASSIGNING <fs_fldcat>.

    CASE <fs_fldcat>-fieldname.
      WHEN 'DOCNUM'.
        <fs_fldcat>-scrtext_l = text-001.
        <fs_fldcat>-scrtext_m = text-001.
        <fs_fldcat>-scrtext_s = text-001.
        <fs_fldcat>-reptext   = text-001.
        <fs_fldcat>-hotspot = 'X'.
        <fs_fldcat>-fix_column = 'X'.
        <fs_fldcat>-key = 'X'.
        <fs_fldcat>-outputlen = 18.

      WHEN 'STATUS' or 'SNDPRN' or 'MESTYP' or 'MESCOD' or 'CREDAT' or 'CRETIM'.

      WHEN 'REFINT' or 'REFGRP' or 'REFMES'.
        <fs_fldcat>-outputlen = 10.

      when 'MAXSEGNUM'.
        <fs_fldcat>-scrtext_l = text-003.
        <fs_fldcat>-scrtext_m = text-003.
        <fs_fldcat>-scrtext_s = text-003.
        <fs_fldcat>-reptext   = text-003.

      when others.
        <fs_fldcat>-no_out = 'X'.

    ENDCASE.
  ENDLOOP.

ENDFORM.                    " field_catalog_merge

1.2.3 ZMM_EDI_GETIDOCSFROMPO_STATO01
*----------------------------------------------------------------------*
***INCLUDE ZMM_EDI_GETIDOCSFROMPO_STATO01 .
*----------------------------------------------------------------------*
*&---------------------------------------------------------------------*
*&      Module  STATUS_2000  OUTPUT
*&---------------------------------------------------------------------*
module STATUS_2000 output.
  SET PF-STATUS '2000'.
  SET TITLEBAR '2000' with gv_po.
endmodule.                 " STATUS_2000  OUTPUT
*&---------------------------------------------------------------------*
*&      Module  init_values_2000  OUTPUT
*&---------------------------------------------------------------------*
module init_values_2000 output.

  data: ls_layout type lvc_s_layo.

  if gref_container is initial.

    create object gref_container
      exporting
        container_name              = c_container
      exceptions
        cntl_error                  = 1
        cntl_system_error           = 2
        create_error                = 3
        lifetime_error              = 4
        lifetime_dynpro_dynpro_link = 5
        others                      = 6.

    create object gref_grid
      exporting
        i_parent          = gref_container
      exceptions
        error_cntl_create = 1
        error_cntl_init   = 2
        error_cntl_link   = 3
        error_dp_create   = 4
        others            = 5.

* Set layout Table Display
*    ls_layout-cwidth_opt = 'X'.
    ls_layout-numc_total = ' '.
    ls_layout-sel_mode   = 'A'.
    ls_layout-zebra      = 'X'.
    ls_layout-totals_bef = ' '.

    gs_variant-report = 'ZMM_EDI_GETIDOCSFROMPO'.

* Program field attributes
    perform field_catalog_merge.

    call method gref_grid->set_table_for_first_display
      exporting
        i_structure_name            = c_itab_alv
        is_variant                  = gs_variant
        is_layout                   = ls_layout
        i_save                      = 'A'
        i_default                   = 'X'
      changing
        it_outtab                   = int_edidc[]
        it_fieldcatalog             = gt_fldcat
   exceptions
     invalid_parameter_combination = 1
     program_error                 = 2
     too_many_lines                = 3
     others                        = 4.

    create object gref_events.

    set handler gref_events->on_link_click   for gref_grid.
    set handler gref_events->handle_double_click   for gref_grid.

  else.

    call method gref_grid->refresh_table_display.

  endif.

endmodule.                 " init_values_2000  OUTPUT

1.2.4 ZMM_EDI_GETIDOCSFROMPO_USERI01
*----------------------------------------------------------------------*
***INCLUDE ZMM_EDI_GETIDOCSFROMPO_USERI01 .
*----------------------------------------------------------------------*
*&---------------------------------------------------------------------*
*&      Module  user_command_2000  INPUT
*&---------------------------------------------------------------------*
MODULE user_command_2000 INPUT.

  CASE okcode.

    WHEN 'BACK' OR 'CANC' OR 'EXIT'.
      IF NOT gref_container IS INITIAL.
        CALL METHOD gref_container->free.
      ENDIF.

      CALL METHOD cl_gui_cfw=>flush.
      LEAVE TO SCREEN 0.

  ENDCASE.

  CLEAR okcode.

ENDMODULE.                 " user_command_2000  INPUT
1.3 GUI Title

1.3.1 2000

IDocs found for PO &

1.4 GUI Status

1.4.1 2000

BACK, EXIT, CANC

1.5 Screens

1.5.1 2000 – ALV Grid Control

1.5.1.1 Screen layout

1.5.1.2 Flow Logic

PROCESS BEFORE OUTPUT.
  MODULE init_values_2000.
  MODULE status_2000.

PROCESS AFTER INPUT.
  MODULE user_command_2000.
1.5.1.3 Element List

1.6 Text Elements

1.6.1 Selection Texts

P_INVOIC
INVOIC (810 - Vendor Invoice)

P_MBGMCR
MBGMCR (856 - Goods Receipt)

P_ORDRSP
ORDRSP (855 - PO Ack)

P_PO
Insight PO Number

P_SHPCON
SHPCON (856 - Post GR- Cr Del)

P_WE09
Show WE09

P_ZEROS
PO in IDoc has no leading zero

SO_DAT
IDoc Creation Date

1.6.2 Text Symbols

001
IDoc Number

003
Found in Segment

BL1
Select PO Number to search for

BL2
Which IDocs do you want to search for?

BL3
IDoc Creation Date range

BL4
Report Controls

ER1
Invalid Insight PO Number:

ER2
Error reading segment:

ER3
IDoc could not be read:

ER4
PO Not found in data! Records read:

I01
Number of records found:

M01
When viewing in WE09 only 1 message type at a time is doable!
1.7 Transaction

1.7.1 ZFINDIDOC
2 Result

2.1 Selection Screen

2.2 Output

